Spreadsheets in Education (eJSiE)

Volume 5 | Issue 1 Article 4

December 2011

Time-Value Concepts, Bond Valuation, and Corresponding Spreadsheet Functions

Yi Feng
Ryerson University, yi.feng@ryerson.ca

Clarence C. Y. Kwan
McMaster University, kwanc@mcmaster.ca

Time-Value Concepts, Bond Valuation, and Corresponding Spreadsheet Functions

Abstract
This paper extends the coverage of bond valuation in introductory finance textbooks. Specifically, it extends the basic model for bond valuation to a more versatile model, which can accommodate valuation dates besides those matching the coupon dates of a bond. Such an extension allows the accrued interest to be properly recognized. Various spreadsheet-based exercises with data from bond quotations are proposed, in order to make the bond model more practically relevant and to enhance the learning experience of students. Connections between some bond-specific spreadsheet functions and the underlying time-value concepts are also established.

Keywords
Time-value concepts, bond valuation, accrued interest, bond quotations

Cover Page Footnote
The authors wish to thank K. Brewer for helpful discussions and the anonymous reviewers for thoughtful comments and suggestions.

Time-Value Concepts, Bond Valuation, and Corresponding Spreadsheet Functions[1]

Yi Feng
Ted Rogers School of Management, Ryerson University
<yi.feng@ryerson.ca>

Clarence C.Y. Kwan
DeGroote School of Business, McMaster University
<kwanc@mcmaster.ca>

December 2011

[1]Kwan is the corresponding author. The authors wish to thank K. Brewer for helpful discussions and the anonymous reviewers for thoughtful comments and suggestions.

Abstract

This paper extends the coverage of bond valuation in introductory finance textbooks. Specifically, it extends the basic model for bond valuation to a more versatile model, which can accommodate valuation dates besides those matching the coupon dates of a bond. Such an extension allows the accrued interest to be properly recognized. Various spreadsheet-based exercises with data from bond quotations are proposed, in order to make the bond model more practically relevant and to enhance the learning experience of students. Connections between some bond-specific spreadsheet functions and the underlying time-value concepts are also established.

Keywords: Time-value concepts, bond valuation, accrued interest, bond quotations

Time-Value Concepts, Bond Valuation, and Corresponding Spreadsheet Functions

1 Introduction

Finance is a core component of business education. As an academic discipline pertaining to financial matters, it provides guidance for asset pricing and business valuation, for decisions by governments, business corporations, and other institutions, and for decisions by individual investors. A foundation that is common for a wide range of issues considered in finance is the time value of money. In a world of compound interest, it is the idea of a current dollar being worth more than a future dollar that provides the basis to guide various financial and related decisions. What relates the current and future values is a discount rate based on the passage of time and risk considerations.

In standard finance textbooks at the introductory level, there are many practical time-value examples that students can readily relate to in daily lives. Periodic repayments of car loans or mortgage loans are just some of such examples. There are also various financial functions in Microsoft Excel^{TM} that can perform directly the corresponding computations. In a recent paper, Sugden and Miller (2011) have presented some complementary approaches by using Excel to deliver basic time-value concepts. The Sugden-Miller paper, which states the present value of some periodic repayments or the future value of some periodic deposits as a geometric series, allows the underlying parameters in financial expressions to be related, thus enhancing the learning experience of students.

As an extension to the Sugden-Miller paper, this paper applies time-value concepts to bond valuation, also with the help of Excel tools.[1] A bond is a traditional long-term financial instrument. If an investor buys a bond, the investor lends money to its issuer, which can be a government, a governmental agency, a corporation, or an institution. In return, the issuer will typically provide equal periodic payments, called coupons, to the bondholder. When the bond eventually matures, the bondholder will also receive a payment that is equal to the face value (par value) of the bond.[2] How much investors are willing to pay for a bond's future cash flows depends on the discount rates involved, which in turn depends on the credit worthiness of the issuer.

[1] Although the spreadsheet illustrations in this paper are Excel-based, the ideas involved are applicable to other spreadsheets as well. For example, the spreadsheet functions and tools in Google Docs^{TM}, Apple Numbers^{TM}, and OpenOffice.org^{TM} can be used for some, if not all, of the illustrations in this paper.

[2] In practice, there are also bonds with non-standard repayment schemes and bonds embedded with various options. However, for the analytical materials to be suitable for introductory finance courses as intended, this paper considers only option-free bonds. The non-standard bond features considered are confined to irregular first and last coupons, if any.

For a given discount rate, known as the yield to maturity, which extends over a long period from the date of the bond issue to the date of its maturity, the present value of the individual coupons and the terminal payment (the face value) can easily be computed. The bond price is simply the sum of such present values. In fact, this is how bond valuation is presented in standard introductory finance textbooks. In practice, when a new bond is issued, the yield to maturity tends to be similar to each coupon as a proportion of the face value of the bond (the coupon rate). Likewise, the bond price tends to be similar to its face value; that is, any premium or discount on the bond tends to be small as intended. However, as investors' required return on bond investment changes over time, so does the bond price.

What complicates matters in bond valuation is that, although a bond is a long-term debt instrument, the bondholder can sell it to others any time before its maturity. The sales price reflects its prevailing yield to maturity in the bond market. According to time-value concepts, regardless of how much time has passed since a bond's existence, its price is always the present value of the remaining coupons and the terminal payment, net of the part of the next coupon, called the accrued interest, that its seller is entitled to receive. Such a price is commonly called the clean price, as opposed to the full price (or dirty price), which includes the accrued interest. Accrued-interest and present-value computations are always based on the settlement date of the transaction, which is the date for payment and ownership transfer; it is usually three business days after the transaction. To complicate matters even further, there are different day-counting conventions for accrued-interest computations in different bond markets.

From an analytical perspective, however, the key difference between the textbook version of bond valuation and a practical version is whether there is a full period between the date of bond valuation and the next coupon date. As students who are taught time-value concepts are expected to be familiar with relating values at different points on the time scale, to relax the full-period requirement from the textbook version should not be analytically burdensome for them. By presenting a practical version of bond valuation, which is still based on familiar time-value concepts, to supplement the coverage in standard textbooks, this paper is intended to make the analytical materials involved more practically relevant and more accessible to students.

2 A Pedagogic Approach and the Roles of Excel

Typical bond quotations, as reported daily in the financial media and shown as examples in finance textbooks, include annual coupons, maturity dates, prices, and annual yields.[3] For a given bond market, what are implicit in the reported data are the annual frequency of coupon payments and the face value of each bond. More detailed quotations also provide bid and ask prices and the corresponding yields, where the bid price is the highest price that potential buyers are willing to pay and the ask price is the lowest price that potential sellers are willing to accept. If unspecified, the quoted prices and yields are typically on the bid side.

In bond quotations, although the coupons and the maturity date of each bond are fixed, its price and yield do vary over time. These changes are results not only of the passage of time, but also of changes in bond market conditions and/or the credit worthiness of the issuer in question. To assist investors in bond investment decisions, Excel has various bond-specific functions. For example, the functions PRICE and YIELD provide the price and the yield to maturity of a bond given the remaining information about the bond, respectively. There are also miscellaneous functions to compute the accrued interest, to perform day counting under different market conventions, and to show coupon dates.[4]

In view of the practical relevance of various bond-specific Excel functions for assisting investment decisions, this paper relates, from a pedagogic perspective, such functions to the corresponding analytical expressions that are based on time-value concepts. To facilitate the derivation of a more versatile model for bond valuation afterwards, this paper starts with the textbook version in Section 3. Though a basic model, the textbook version still allows some relevant bond properties to be deduced. It also allows some algebraic relationships among the bond price and its various underlying parameters to be established. Such relationships, in turn, allow each parameter from the bond price formula to be solved, analytically or numerically, in terms of the remaining parameters including the bond price.

To see why the above-mentioned exercises can enhance students' learning experience of the textbook materials on bond valuation, notice that each coupon bond is analytically equivalent to a combination of two debt components. One component pertains to the periodic coupons only;

[3] See, for example, Bodie, Kane, and Marcus (2004, Chapter 9), Ross et al. (2011, Chapter 6), and Ross et al. (2010, Chapter 7) for some actual bond quotations.

[4] There are also Excel functions to compute two common versions of bond duration, as intended to capture the bond-price sensitivity in response to yield changes. The topic of duration is typically covered in some detail in investment textbooks. As this paper is intended to cover bond topics that are relevant to students in introductory finance courses, the topic of duration is omitted here.

the remaining component is a pure discount bond with its only payment being equal to the face value of the original coupon bond. The two components that form a coupon bond have a common maturity date and a common yield to maturity each period. The component based on the periodic coupons have all analytical features of a loan with equal periodic repayments.

When time-value concepts are taught to students, simple exercises involving a loan with equal periodic repayments are often formulated with the size of the loan or the amount of each repayment being the unknown. More advanced exercises are to find the number of repayments to fulfill the loan obligation or to find the underlying interest rate instead. To solve the number of repayments analytically requires the use of logarithm in algebraic manipulations. The interest rate of a loan can only be solved numerically, with the exception of special cases for which analytical solutions are possible. From our experience as instructors of introductory finance courses, such exercises are indeed valuable in helping students understand the underlying time-value concepts. Thus, the same can be expected from analogous exercises involving coupon bonds.

Although the textbook version of bond valuation is analytically simple, computational exercises analogous to those for loans with equal periodic repayments, if performed manually, can still be tedious and distracting. It is worth noting that many business students have already had some hands-on experience with Excel by the time they enroll in introductory finance courses. Thus, if Excel is utilized for computational exercises, students can pay more attention to the analytical and conceptual issues involved. In fact, some textbooks have explicitly recommended the use of Excel for various finance topics.[5]

To make the Excel illustration in this paper practically relevant, an actual bond quotation that fits the textbook version of bond valuation is used. Suppose that a piece of information in the bond quotation is intentionally concealed. The missing item can be the price, the yield to maturity, the coupon, or the maturity date of the bond. Of interest is how the missing information can be deduced. Although the use of some bond-specific Excel functions can provide the answers directly, it is the connection between the Excel results and the underlying analytical materials that is of pedagogic importance.

Section 4 extends the textbook version of bond valuation by relaxing the requirement of a full period between the date of bond valuation and the next coupon date. The approach, which is still based on time-value concepts and algebraic tools that are familiar to business students, allows the revised bond price formula to retain the intuitive appeal of the textbook version. The revised

[5] The introductory finance textbooks by Ross et al. (2011) and Ross et al. (2010) are notable examples.

bond price formula, in turn, allows various underlying parameters to be solved, analytically or numerically, in terms of the remaining parameters, in ways analogous to those in Section 3. As one of such parameters pertains to the time interval for the accrued-interest computation, the settlement date can be deduced from the remaining information in a bond quotation as well. Likewise, Excel is utilized for the corresponding numerical illustration.

Once the foundation for practical bond valuation has been established in Section 4, we are ready to relate pedagogically in Section 5 the various bond-specific Excel functions to the corresponding analytical expressions that are based on time-value concepts. The results from such functions are compared with those in the corresponding illustrative example in Section 4. As expected, the results are consistent. Section 5 also addresses various practical issues in bond valuation. Finally, some concluding remarks are provided in Section 6.

3 A Basic Model for Bond Valuation

Let us start with the simplest version of a coupon bond. The bond has a face value of F and pays a constant coupon of C once a year until maturity, which is n years from now. Suppose that the next coupon will be exactly one year from now and that the investor will receive, in addition, the face value when the bond matures. Suppose also that the yield to maturity, expressed as an annual rate of return over the remaining life of the bond, is r. According to time-value concepts, the current value of the bond is

$$P = \frac{C}{1+r} + \frac{C}{(1+r)^2} + \cdots + \frac{C}{(1+r)^n} + \frac{F}{(1+r)^n}. \tag{1}$$

In practice, however, although only some bonds pay coupons annually, most bonds have semi-annual coupons. There are also bonds with monthly or quarterly coupons. Nevertheless, equation (1) can accommodate bonds with higher frequencies of coupon payments. For a bond that pays coupons m times a year, where m can be 1, 2, 4, or 12, each coupon payment period (or, simply, each period) is $12/m$ months. Accordingly, the symbol C in equation (1) will represent instead the amount of each coupon payment, with the annual amount being mC; the symbol n will represent instead the number of remaining periods until the bond matures; and the symbol r will represent instead investors' required return each period, over the remaining life of the bond. For bond quotations in practice, the coupon and the yield to maturity are typically stated as annual figures of mC and mr, respectively.

Analytically, the first n terms on the right hand side of equation (1) form a finite geometric series, which is a series with a common ratio of adjacent terms throughout. Here, the common ratio of term $i+1$ and term i is $1/(1+r)$, for $i = 1, 2, \ldots, n-1$. As shown below, this analytical feature allows us to write equation (1) in a computationally convenient form when n is large. The analytical materials involved are set at a level similar to that in the paper on the mathematics of bond prices by Lawrence and Shankar (2007).

By letting

$$A = \frac{1}{1+r} + \frac{1}{(1+r)^2} + \cdots + \frac{1}{(1+r)^n}, \tag{2}$$

which is commonly called the present value of annuity factor, we also have

$$(1+r)A = 1 + \frac{1}{1+r} + \frac{1}{(1+r)^2} + \cdots + \frac{1}{(1+r)^{n-1}}. \tag{3}$$

Differencing the two expressions above leads to

$$A = \frac{1-(1+r)^{-n}}{r}. \tag{4}$$

Thus, equation (1) reduces to

$$P = C\left[\frac{1-(1+r)^{-n}}{r}\right] + \frac{F}{(1+r)^n}. \tag{5}$$

Notice that equation (5) can accommodate all cases of $n \geq 1$; for the case of $n = 1$, the factor $[1-(1+r)^{-n}]/r$ reduces to $1/(1+r)$, as implied by equation (1).

3.1 Essential Bond Properties

Equation (5) provides an algebraic relationship among P, C, F, n, and r. It is this relationship that allows some fundamental properties in bond valuation to be established. The most well-known bond property, which is readily noticeable in equation (1), is the inverse relationship between price and yield. That is, an increase (a decrease) in r corresponds to a decrease (an increase) in P.

By subtracting F from both sides of equation (5), we have

$$\begin{aligned} P - F &= C\left[\frac{1-(1+r)^{-n}}{r}\right] + \frac{F}{(1+r)^n} - F \\ &= F\left(\frac{C}{F} - r\right)\left[\frac{1-(1+r)^{-n}}{r}\right]. \end{aligned} \tag{6}$$

With $P - F$ being the premium or discount of a bond, depending on its sign, and with C/F, the coupon as a proportion of the bond's face value, being the coupon rate, equation (6) reveals the following: A bond will sell at a premium (discount) if the coupon rate is above (below) investors' required return for holding the bond. A bond will sell at par, with $P = F$, if the coupon rate matches investors' yield requirement.

Equation (6) reveals a further bond property. Suppose that there is no change in the yield to maturity as a bond approaches its maturity date. As the present value of annuity factor that $[1-(1+r)^{-n}]/r$ represents decreases with a decreasing n, the magnitude of the premium or discount of the bond that $P-F$ represents decreases as well. This property captures the idea that, the shorter (longer) it takes for the bond to mature, the less (greater) is the impact on the bond price by the difference between the coupon rate and the bond yield.

3.2 Some Exercises to Facilitate Learning

From a pedagogic perspective, equation (5) can be used for various exercises to facilitate learning. As any of P, C, F, n, and r can be treated as an unknown, the algebraic relationship that the equation provides is useful for solving the unknown. Notice that, being a specific round number, F can easily be inferred from P, C, and r without using equation (5). For example, in the U.S., F tends to be \$500 for municipal bonds, \$1,000 for corporate bonds, and \$10,000 for federal government bonds. Further, in bond quotations, bond prices and coupons are typically scaled for $F = \$100$.

To solve equation (5) for P or C being the only unknown is straightforward. Now, suppose that the only unknown is n instead. It can be solved only if the bond sells at a premium or at a discount. Otherwise, with $P = F$ and $C/F = r$, equation (5) is always satisfied regardless of the value of n. For cases where $P \neq F$ and $C/F \neq r$, we can write the equation as

$$(1+r)^n = \frac{rF - C}{rP - C}. \tag{7}$$

Taking the natural logarithm of both sides of equation (7) leads to

$$n = \left[\ln\left(\frac{rF - C}{rP - C}\right)\right] \Big/ \ln(1+r), \tag{8}$$

which allows n to be determined directly.

In general, given P, C, F, and n, the unknown r cannot be solved analytically and thus a numerical approach is required for its determination. Although there can be as many as n different values of r that satisfy equation (5), including complex values, the only value that makes economic sense must be real and strictly positive. If $r = 0$, no discount will apply to the series of n coupons and the face value of the bond paid at maturity. If so, the bond will sell at $P = nC + F$, its highest possible price. As r increases, with C, F, and n held constant, P will decline monotonically; there will be a specific value of r that allows the price according to equation (5) to match the given P. Given this inverse relationship between P and r, a numerical search for r can easily be performed with Excel.

3.3 An Excel Example

Figure 1 shows an Excel example based on a Canadian federal government bond, with data accessed from the GlobeInvestor Gold website of The Globe and Mail, a Canadian national newspaper.[6] The example considers a semi-annual coupon bond that matures in 15/03/2021. Here, the date convention of *dd/mm/yyyy* is followed. In order to fit the bond data to the basic model above, a 15/03/2011 (Tuesday) settlement date is selected. As the settlement date is unavailable from this data source, a three-business-day settlement is assumed; that is, the bond quotation of 10/03/2011 (Thursday a week earlier) is used for the example. All data from the bond quotation are shaded in Figure 1. The price and yield data used for the example are on the bid side.

For this semi-annual coupon bond with a face value of $F = \$100$, where the annual coupon frequency is $m = 2$, the quoted \$10.50 annual coupon and 3.39% annual yield to maturity correspond to a semi-annual coupon rate of $C/F = 0.1050/m = 0.05250$ and a semi-annual yield to maturity of $r = 0.0339/m = 0.01695$, respectively. With the coupon rate being higher than the yield, the quoted bond price $P = \$159.87$ corresponds to a \$59.87 premium over the bond's face value. Based on the 15/03/2011 settlement date and the 15/03/2021 maturity date, there are 20 remaining semi-annual coupons, with each being \$5.25.[7]

In Figure 1, all computed values are shown in column B, with individual labels and cell formulas provided in columns A and D at adjacent locations. Notice that, although some computations can also be performed directly by using bond-specific Excel functions, including those functions for computing the bond price and the yield, their use will be postponed until Section 5. The postponement will allow the analytical focus here to be maintained more effectively.

We now illustrate that, under the condition of a full period between the settlement date and the next coupon date, if either the price P or the annual coupon mC is concealed from the bond quotation, it can be deduced from the simple model in equation (5). The cell formulas for the computations of P and C in terms of the remaining data in each case, based on equation (5), are as provided in B16:B17. Except for minor rounding errors, the computed P and mC (in B18) are consistent with the quoted data in B5 and B8, respectively, as expected.

[6] Data access from the website http://gold.globeinvestor.com/ is subscription-based.

[7] The number of semi-annual periods before maturity in the example, which is $n = 20$, as stored in B12, is so obvious that any computation for n seems redundant. However, to accommodate less obvious cases, the cell formula for B12, which is =ROUND((B4-B3)*B7/365.25,0), uses the number of days between settlement date and the maturity date to obtain an approximate number of semi-annual periods and then to reach the nearest integer by rounding the result thus obtained. The zero as indicated in the Excel function ROUND is to specify that no decimal places be shown for the rounded result. The use of a 365.25-day year is to account for the extra day in a leap year once every four years.

	A	B	C	D	E	F	G	H
1	A Canadian Federal Government Bond							
2								
3	Settlement date	15/03/2011						
4	Maturity date	15/03/2021						
5	P: price	$159.87						
6	F: face value	$100						
7	m: annual coupon frequency	2						
8	mC: annual coupons	$10.50						
9	C: coupon each period	$5.25		=B8/B7				
10	mr: annual yield to maturity	0.0339						
11	r: yield to maturity, per period	0.01695		=B10/B7				
12	n: # of periods until maturity	20		=ROUND((B4-B3)*B7/365.25,0)				
13								
14	To find P, C, or n:							
15								
16	P as the only unknown	$159.8772		=B9*((1-(1+B11)^(-B12))/B11)+B6*(1+B11)^(-B12)				
17	C as the only unknown	$5.2496		=(B5-B6*(1+B11)^(-B12))*B11/(1-(1+B11)^(-B12))				
18	mC	$10.4991		=B7*B17				
19	n as the only unknown	20		=ROUND(LN((B11*B6-B9)/(B11*B5-B9))/LN(1+B11),0)				
20	Deduced maturity date	15/03/2021		=B3+ROUND(B19*365.25/B7,0)				
21								
22	To find r, the only unknown, with Goal Seek:							
23				Set cell: B29				
24	initial r	0.000000		To value: 0				
25	r	0.016953		By changing cell: B25				
26	mr	0.033906		=B7*B25				
27	Computed P	$159.8700		=IF(B25=0,B12*B9+B6,				
28				B9*((1-(1+B25)^(-B12))/B25)+B6*(1+B25)^(-B12))				
29	Computed P - given P	$0.0000		=B27-B5				
30								
31	To find r, the only unknown, with Solver:							
32				Set target cell: B39				
33				Equal to value of: 0				
34	initial r	0.000000		By changing cell: B35				
35	r	0.016953		Subject to constraint: B35>=0				
36	mr	0.033906		=B7*B35				
37	Computed P	$159.8700		=IF(B35=0,B12*B9+B6,				
38				B9*((1-(1+B35)^(-B12))/B35)+B6*(1+B35)^(-B12))				
39	Computed P - given P	$0.0000		=B37-B5				
40								

Figure 1 An Excel Example Based on a Basic Model for Bond Valuation, with Information from a Quotation of a Canadian Federal Government Bond.

We proceed to illustrate that, if the maturity date in the bond quotation is concealed, it can be deduced by determining n first. To compute n in B19, given the remaining data, requires equation (8) instead. As n is intended to be an integer, the use of the function ROUND in the cell formula for B19 ensures that the computed result from equation (8) be rounded to become an integer. Given $n = 20$, $m = 2$, and the 15/03/2011 settlement date, the cell formula for B20, =B3+ROUND(B19*365.25/B7,0), provides the maturity date of 15/03/2021, as expected.[8]

If the annual yield mr in the bond quotation is concealed, we provide two alternative approaches to find r given the remaining data. One approach is via Goal Seek, which can be accessed from What-If Analysis under the Excel menu Data. This requires P to be computed in B27 first for an initial r in B25 (copied from B24), along with the given values of C, F, and n. Given the inverse relationship between P and r, for $r \geq 0$, the initial r can be any non-negative number. In this illustration, we let the initial r be zero, which corresponds to the initial P being $nC + F$. The difference between the computed P and the given P is placed in B29. Goal Seek allows the value of r in B25 to vary until the difference as shown in B29 is completely diminished. The Goal Seek result of $r = 0.016953$, as shown in B25, is consistent with its given value in B11. That is, the corresponding $mr = 0.033906$ is consistent with the annual yield to maturity in the bond quotation.

The other approach is via Solver, which is also under the Excel menu Data.[9] Likewise, the difference between the P computed in B37 and the given P for the same initial r in B35 (copied from B34), along with the given values of C, F, and n, are stored in B39. The Solver result is achieved by changing the value of r in B35, subject to the constraint that it must be non-negative, until the difference as shown in B39 is completely diminished. As shown in B35, the Solver result is also $r = 0.016953$, corresponding to $mr = 0.033906$, which is consistent with the annual yield to maturity in the bond quotation.

4 A More Versatile Model for Bond Valuation

For bond valuation on dates other than the start of each period, equation (5) has to be revised. A practical reason is that, if an investor who owns a bond sells it with the settlement date of the transaction prior to the next coupon date, the investor is entitled to receive part of the next

[8] If the 15/03/2021 maturity date instead of the settlement date is provided, the latter can be deduced by using the cell formula =B4-ROUND(B19*365.25/B7,0).

[9] The use of both Goal Seek and Solver in the same illustration is intended to help students understand better the analytical concepts involved, without the encumbrance of numerical search details. See the website of the Science Education Resource Center at Carleton College (http://serc.carleton.edu/econ/spreadsheets/do.html) for illustrations of these numerical tools in teaching economics. See also Cahill and Kosicki (2001) for discussions of black box issues associated with spreadsheets.

coupon from the buyer on the settlement date. Here, the settlement date is the date when the buyer pays and the seller delivers the bond certificate; it is typically three business days following the transaction. The part of the coupon that the seller receives is called the accrued interest. It is based on the number of days from the previous coupon date to the settlement date, as a proportion of the number of days from the previous coupon date to the next one. Specifically, if the proportion is α, the accrued interest will be αC in practice.

To complicate matters in practice, there are different day-counting conventions for individual bond markets.[10] Accordingly, there are different ways to determine α, depending on the markets involved. From a pedagogic perspective, however, such differences are of secondary importance, as they do not affect the way the model for bond valuation is formulated. Thus, for now, let us ignore any existing differences in day-counting conventions and consider each settlement date between two adjacent coupon dates to correspond to a specific α, with $0 \leq \alpha < 1$, so that time-value concepts can still be applied directly.[11]

Suppose that the required return each period is r until the bond matures. According to time-value concepts, the value of the bond that has n remaining coupon payments until maturity, on the first of such payment dates, is

$$P^* = C + \frac{C}{1+r} + \frac{C}{(1+r)^2} + \cdots + \frac{C}{(1+r)^{n-1}} + \frac{F}{(1+r)^{n-1}}. \tag{9}$$

The discount factor for both the final coupon C and the face value F is $1/(1+r)^{n-1}$ because they are to be received at the end of $n-1$ periods, as of that coupon date. With the accrued interest αC accounted for, the total purchase price of the bond on the settlement date, which is the proportion $(1-\alpha)$ of a period prior to the first coupon date, is

$$P + \alpha C = \frac{P^*}{(1+r)^{1-\alpha}}. \tag{10}$$

The idea here is to equate the total purchase price to the present value of P^*, as of the settlement date. Combining the above two equations leads to

$$P = \frac{1}{(1+r)^{1-\alpha}} \left[C + \frac{C}{1+r} + \frac{C}{(1+r)^2} + \cdots + \frac{C}{(1+r)^{n-1}} + \frac{F}{(1+r)^{n-1}} \right] - \alpha C. \tag{11}$$

[10] See, for example, Christie (2003) and the undated document "Canadian Conventions in Fixed Income Markets: A Reference Document of Fixed Income Securities Formulas and Practices, Release 1.1," which can be accessed from the website of Investment Industry Association of Canada (http://www.iiac.ca/original_documents/Canadian%20Conventions%20in%20FI%20Markets%20-%20Release%201.1.pdf).

[11] The case of $\alpha = 1$ is when the settlement date falls on a coupon date. In such a situation, the seller of the bond will receive directly from the issuer the full amount of the coupon involved. In the absence of any accrued interest, the clean price and the full price must be the same on that date. Thus, the case of $\alpha = 1$ is irrelevant in practice.

In practice, the price P of a bond in equation (11) is called a clean price, as opposed to a full price, which is $P+\alpha C$ instead. The clean price of a bond is commonly used for bond quotations in practice, as it allows any premium or discount of the bond to be identified immediately. Observed changes in the clean price allow investors to assess whether the underlying changes in investors' required return for holding the bond are market-wide or issuer-specific in nature. Should a full price exceeding the face value be quoted instead, investors would have to compute the accrued interest before knowing whether a premium existed. Likewise, should a quoted full price be below the face value, investors would also have to compute the accrued interest before knowing the severity of the discount involved. In either case, interpretations of observed price changes would inevitably be less direct.

We can write equation (11) equivalently as

$$P = (1+r)^{\alpha}\left[\frac{C}{1+r} + \frac{C}{(1+r)^2} + \cdots + \frac{C}{(1+r)^n} + \frac{F}{(1+r)^n}\right] - \alpha C. \tag{12}$$

To give equation (12) an intuitive interpretation, notice that the sum of the bracketed terms on the right hand side of the equation can be viewed as the bond value, at a point on the time scale, when the bond has exactly n full periods before maturity. Notice also that the future-value factor $(1+r)^{\alpha}$ there moves the bond value forward, from that point to the settlement date. As the move shortens the waiting times for the investor to receive the n future coupons and the face value of the bond at maturity, it has a positive effect on the bond value. In contrast, the accrued interest αC provides instead a negative effect of the move.

In view of the equivalence of equations (1) and (5), we can also write equation (12) as

$$P = (1+r)^{\alpha}\left\{C\left[\frac{1-(1+r)^{-n}}{r}\right] + \frac{F}{(1+r)^n}\right\} - \alpha C \tag{13}$$

for computational convenience. If $\alpha = 0$, equation (13) reduces to equation (5). That is, the model here encompasses the basic model in Section 3 as a special case.

4.1 A Closer Look at Premium and Discount

According to the basic model for bond valuation in Section 3, a bond will sell at a premium, with $P > F$, if the coupon rate C/F exceeds investors' required return r each period over the remaining life of the bond. Likewise, the bond will sell at a discount, with $P < F$, if $C/F < r$, and it will sell at par, with $P = F$, if $C/F = r$. For the model based on equation (11), if $r = 0$, as the corresponding price P will be $(n-\alpha)C + F$, which exceeds F, the bond will inevitably sell at a premium. Given the inverse relationship between P and r that equation (11) provides, an increase in r from $r = 0$ will correspond to a decrease in P and will eventually result in $P < F$ if r is sufficiently large.

For the model based on equation (11), if $C/F = r$, will the bond still sell at par, with $P = F$? To examine this analytically, let us rewrite equation (13) as

$$(P + \alpha C)(1+r)^{-\alpha} = C\left[\frac{1-(1+r)^{-n}}{r}\right] + \frac{F}{(1+r)^n}. \tag{14}$$

Subtracting F from both sides of equation (14) leads to

$$\begin{aligned}(P + \alpha C)(1+r)^{-\alpha} - F &= C\left[\frac{1-(1+r)^{-n}}{r}\right] + \frac{F}{(1+r)^n} - F \\ &= F\left(\frac{C}{F} - r\right)\left[\frac{1-(1+r)^{-n}}{r}\right]. \end{aligned} \tag{15}$$

If $C/F = r$, we have

$$P + \alpha C = F(1+r)^{\alpha} \tag{16}$$

or, equivalently,

$$\frac{P}{F} = (1+r)^{\alpha} - \alpha r. \tag{17}$$

For $\alpha = 0$, as $(1+r)^{\alpha} - \alpha r = 1$, equation (17) leads to $P = F$. For $0 < \alpha < 1$, in contrast, as $(1+r)^{\alpha} - \alpha r < 1$, P will be strictly less than F, resulting in a small discount in the bond price.[12]

The correspondence between $C/F = r$ and $P = F$ at each coupon date is as expected. This is because the model here actually encompasses the basic model for bond valuation as a special case. For other dates before maturity, where $0 < \alpha < 1$, the accrued interest αC is computed according to a simple-interest scheme in practice. The small discount in the bond price, which pertains to $C/F = r$, captures indirectly the difference that exists between simple-interest and compound-interest approaches in establishing one of the components of bond valuation.

4.2 More Exercises to Facilitate Learning

We can use equation (13) to solve for any unknown among P, C, F, n, r, and α in terms of the rest. As indicated earlier, F can easily be deduced by inspecting the remaining data without relying on any computations. If P is the unknown, equation (13) can be used directly. If C is the unknown

[12] The analytical property of $(1+r)^{\alpha} - \alpha r < 1$, for $r > 0$ and $0 < \alpha < 1$, can be illustrated numerically with Excel. To verify this analytically, let $y = (1+r)^{\alpha} - 1$, defined over $r > 0$ and $0 \leq \alpha \leq 1$. As $dy/d\alpha = [\ln(1+r)](1+r)^{\alpha} > 0$ and $d^2y/d\alpha^2 = [\ln(1+r)]^2(1+r)^{\alpha} > 0$, y increases with α at an increasing rate. For $0 < \alpha < 1$, the graph of y as a function of α lies below the linear graph of αr, also as a function of α; the two graphs meet at $\alpha = 0$ and $\alpha = 1$. Accordingly, for $0 < \alpha < 1$, we always have $(1+r)^{\alpha} - 1 - \alpha r < 0$ or, equivalently, $(1+r)^{\alpha} - \alpha r < 1$.

instead, then the equation can be rewritten as

$$P(1+r)^{-\alpha} = C\left[\frac{1-(1+r)^{-n}}{r}\right] + \frac{F}{(1+r)^n} - \alpha(1+r)^{-\alpha}C, \tag{18}$$

from which we have

$$C = \frac{P(1+r)^{-\alpha} - F(1+r)^{-n}}{[1-(1+r)^{-n}]/r - \alpha(1+r)^{-\alpha}}. \tag{19}$$

To determine n in terms of P, C, F, r, and α, we can rewrite equation (18) as

$$r(P+\alpha C)(1+r)^{-\alpha} - C = (rF - C)(1+r)^{-n}. \tag{20}$$

It follows that

$$(1+r)^n = \frac{rF - C}{r(P+\alpha C)(1+r)^{-\alpha} - C}. \tag{21}$$

By taking the natural logarithm of both sides of equation (21), we have

$$n = \left\{\ln\left[\frac{rF - C}{r(P+\alpha C)(1+r)^{-\alpha} - C}\right]\right\} \Big/ \ln(1+r). \tag{22}$$

For equation (22) to work as intended, neither $rF - C$ nor $r(P+\alpha C)(1+r)^{-\alpha} - C$ can be zero. Further, if $\alpha = 0$, equation (22) reduces to equation (8), which corresponds to the basic model in Section 3.

In general, if either r or α is treated as an unknown, it cannot be solved analytically in terms of the remaining parameters in equation (13). In the case of r, given the inverse relationship between P and r, for $r > 0$, solving for r numerically with Excel's Goal Seek or Solver is a relatively easy task. Likewise, equation (13) can also be used to solve for α in a similar manner. If r remains the same as time goes by, from the start to the end of a period, while α increases, the magnitude of any premium or discount of the bond will decrease monotonically. It is this feature that makes the numerical search of α with Excel's Goal Seek or Solver easy to implement.

4.3 Another Excel Example

Figure 2 shows an Excel example based on the same Canadian federal government bond in Figure 1. As the day-counting convention for the Canadian bond market is based on actual calendar days, we can avoid digressions arising from day-counting issues at this stage of the Excel illustration. The settlement date for the example is set at 19/05/2011 (Thursday). Under the assumption of a three-business-day settlement, it corresponds to the bond quotation of 16/05/2011 (Monday). Again, all data from the bond quotation are shaded in Figure 2. The price and yield data used for

	A	B	C	D	E	F	G	H
1	A Canadian Federal Government Bond							
2								
3	Settlement date	19/05/2011						
4	Maturity date	15/03/2021						
5	P: price	$161.22						
6	F: face value	$100						
7	m: annual coupon frequency	2						
8	mC: annual coupons	$10.50						
9	C: coupon each period	$5.25		=B8/B7				
10	n: # of periods until maturity	20		=ROUND((B4-B3)*B7/365.25,0)				
11	Lastest coupon date	15/03/2011		=B4-ROUND(B10/B7*365.25,0)				
12	Next coupon date	15/09/2011		=EDATE(B11,12/B7)				
13	mr: annual yield to maturity	0.0319						
14	r: yield to maturity per period	0.01595		=B13/B7				
15	alpha: proportion of period	0.3533		=(B3-B11)/(B12-B11)				
16								
17	To find P, C, or n:							
18								
19	P as the only unknown	$161.2221		=((1+B14)^B15)*(B9*((1-(1+B14)^(-B10))/B14)				
20				+B6*(1+B14)^(-B10))-B15*B9				
21	C as the only unknown	$5.2499		=(B5*(1+B14)^(-B15)-B6*(1+B14)^(-B10))/				
22				(((1-(1+B14)^(-B10)))/B14-B15*(1+B14)^(-B15))				
23	n as the only unknown	20		=ROUND(LN((B14*B6-B9)/(B14*(B5+B15*B9)*				
24				(1+B14)^(-B15)-B9))/LN(1+B14),0)				
25	Deduced maturity date	15/03/2021		=B3+ROUND((B23-B15)*365.25/2,0)				
26								
27	To find r, the only unknown, with Goal Seek:							
28								
29	initial r	0.00000		Set cell: B36				
30				To value: 0				
31	r	0.01595		By changing cell: B31				
32	mr	0.03190		=B7*B31				
33								
34	Computed P	$161.2207		=IF(B31=0,(B10-B15)*B9+B6,((1+B31)^B15)*(B9*((1-				
35				(1+B31)^(-B10))/B31)+B6*(1+B31)^(-B10))-B15*B9)				
36	Computed P - given P	$0.0007		=B34-B5				
37								
38	To find r, the only unknown, with Solver:							
39				Set target cell: B48				
40				Equal to value of: 0				
41	initial r	0.00000		By changing cell: B43				
42				Subject to constraint: B43>=0				
43	r	0.01595						
44	mr	0.03190		=B7*B43				
45								

Figure 2 An Excel Example Based on a More Versatile Model for Bond Valuation, with Information from a Quotation of a Canadian Federal Government Bond.

	A	B	C	D	E	F	G	H
46	Computed P	$161.2200		=IF(B43=0,(B10-B15)*B9+B6,((1+B43)^B15)*(B9*((1-				
47				(1+B43)^(-B10))/B43)+B6*(1+B43)^(-B10))-B15*B9)				
48	Computed P - given P	$0.0000		=B46-B5				
49								
50	To find alpha, the only unknown with Goal Seek:							
51								
52	initial alpha	0.0000		Set cell: B58				
53				To value: 0				
54	alpha	0.35396		By changing cell: B54				
55	Deduced settlement date	19/05/2011		=B4-ROUND((B10-B54)/B7*365.25,0)				
56	Computed P	$161.2202		=((1+B14)^B54)*(B9*((1-(1+B14)^(-B10))/B14)+				
57				B6*(1+B14)^(-B10))-B54*B9				
58	Computed P - given P	$0.0002		=B56-B5				
59								
60	To find alpha, the only unknown, with Solver:							
61				Set target cell: B68				
62	initial alpha	0.00000		Equal to value of: 0				
63				By changing cell: B64				
64	alpha	0.35404		Subject to constraints: B64>=0, B64<=1				
65	Deduced settlement date	19/05/2011		=B4-ROUND((B10-B64)/B7*365.25,0)				
66	Computed P	$161.2200		=((1+B14)^B64)*(B9*((1-(1+B14)^(-B10))/B14)+				
67				B6*(1+B14)^(-B10))-B64*B9				
68	Computed P - given P	$0.0000		=B66-B5				
69								
70	To use bond-specific Excel functions directly:							
71								
72	PRICE	161.2221		=PRICE(B3,B4,B8/B6,B13,B6,B7,1)				
73	YIELD	0.03190		=YIELD(B3,B4,B8/B6,B5,B6,B7,1)				
74	COUPDAYS	184		=COUPDAYS(B3,B4,B7,1)				
75	COUPDAYBS	65		=COUPDAYBS(B3,B4,B7,1)				
76	COUPDAYSNC	119		=COUPDAYSNC(B3,B4,B7,1)				
77	COUPNCD	15/09/2011		=COUPNCD(B3,B4,B7,1)				
78	COUPPCD	15/03/2011		=COUPPCD(B3,B4,B7,1)				
79	COUPNUM	20		=COUPNUM(B3,B4,B7,1)				
80	ACCRINT	1.8546		=ACCRINT(B78,B77,B3,B8/B6,B6,B7,1)				
81								
82	To find the annual coupon via GOAL SEEK and the Excel function PRICE:							
83								
84	initial annual coupon	$3.0000		Set cell: B88				
85	annual coupon	$10.4998		To value: 0				
86				By changing cell: B85				
87	Computed P (using PRICE)	$161.2200		=PRICE(B3,B4,B85/B6,B13,B6,B7,1)				
88	Computed P - given P	$0.00		=B87-B5				
89								

Figure 2 An Excel Example Based on a More Versatile Model for Bond Valuation, with Information from a Quotation of a Canadian Federal Government Bond (Continued).

the example are on the bid side. The computational results and the corresponding cell formulas are displayed in the same manner as those in Figure 1.

From the bond quotation for the 19/05/2011 settlement date, we have a clean price of $P =$ \$161.22 and an annual yield of $mr = 0.0319$, where $m = 2$ is the annual frequency of coupon payments. Accordingly, we have a semi-annual yield of $r = 0.0319/m = 0.01595$. The previous and next coupon dates, as implied by the bond quotation, are 15/03/2011 and 15/09/2011, respectively. Such information, when combined with the remaining data from the bond quotation, allows the number of periods n before maturity and the proportion α of a period for computing the accrued interest to be deduced, as shown in B10 and B15, respectively.

If any of P, C, n, and r is the concealed information in the bond-price formula in equation (13), its determination via an appropriate equation among equations (13), (19), and (22), either directly or with Goal Seek or Solver involved, is similar to that in Figure 1. Thus, the displayed information in rows 17 to 48 of Figure 2 is self-explanatory. Notice that, in the context of a bond quotation, when n is the unknown parameter to be solved, the concealed information in the quotation can be the maturity date or the settlement date. In the example, the maturity date is considered to be the missing information; it is determined with the cell formula for B25, which is =B3+ROUND((B23-B15)*365.25/2,0).[13]

The numerical search for α as the unknown parameter, as shown in rows 50 to 68, is similar to that for r, although the Solver search requires two constraints, which are $\alpha \geq 0$ and $\alpha \leq 1$. Strictly speaking, the correct value of α must correspond to the quotient of two integers, with the numerator being the number of days since the previous coupon date and the denominator being the number of days in a period. Although the rounding error in the quoted yield tends to cause the numerical search result of α to deviate from the accrued interest per dollar of coupon, a correct α as shown in B15 can still be replicated by using the deduced settlement date in B55 and B65.

5 Bond Quotations and Bond-Specific Excel Functions

Excel has various bond-specific functions that can be used to replicate information in bond quotations or to deduce additional information. The functions PRICE and YIELD, which pertain to the former case, allow the clean price and the annual yield to maturity of a bond to be determined if information on one of them is provided. Examples in the latter case include the following: The functions COUPPCD and COUPNCD provide the previous and next coupon dates, respec-

[13] If the settlement date is missing instead, the corresponding cell formula for its determination will be =B4-ROUND((B23-B15)*365.25/2,0).

tively. The functions COUPDAYS, COUPDAYBS, and COUPDAYSNC compute, for a given day-counting convention, the numbers of days in the coupon period that contains the settlement date, between the previous coupon date and the settlement date, and between the settlement date and the next coupon date, respectively. Further, the function COUPNUM provides the number of remaining coupons, and the function ACCRINT shows explicitly the accrued interest.

For the same example in Figure 2, the above bond-specific functions are placed in B72:B80. Specifically, the cell formulas =PRICE(B3,B4,B8/B6,B13,B6,B7,1) and =YIELD(B3,B4,B8/B6,B5, B6,B7,1) provide a clean price of $161.2221 and an annual yield of 0.03190, respectively, matching very well the values in B5 and B13 from the bond quotation. The first six of the seven arguments of each function, which show the corresponding cells containing the required data in a specific order, are self-explanatory. The last argument, which is 1 in the example, pertains to the day-counting convention based on actual calendar days.

The cell formulas =COUPDAYS(B3,B4,B7,1), =COUPDAYBS(B3,B4,B7,1), and =COUP-DAYSNC(B3,B4,B7,1) return 184, 65, and 119 days, respectively. Accordingly, the proportion of a period elapsed is $\alpha = 65/184 = 0.3533$, which matches the value in B15. The cell formulas =COUPNCD(B3,B4,B7,1) and =COUPPCD(B3,B4,B7,1) return the following dates, as expected: 15/09/2011 and 15/03/2011. Further, the cell formulas =COUPNUM(B3,B4,B7,1) and =ACCRINT(B78,B77,B3,B8/B6,B6,B7,1) return the integer 20 and the amount of $1.8546. The integer is consistent with that in B10. The dollar amount, which is the accrued interest, is the same as $\alpha C = 0.3533(\$10.50/2) = \$1.8546$. Again, the final argument of each function, which is 1, indicates the use of actual calendar days for day counting.

Although Excel does not have a built-in function to replicate the annual coupon from a bond quotation, it can easily be deduced via some other bond-specific functions. In Figure 2, we arbitrarily initialize an annual coupon and place it in B85. The cell formula for B87, which is =PRICE(B3,B4,B85/B6,B13,B6,B7,1), provides the corresponding clean price. We then use Goal Seek to vary B85 until there is no numerical difference between the two prices in B87 and B5. As expected, the end result is consistent with the annual coupon in B8.

It is obvious that the use of the above bond-specific functions instead will greatly simplify the computational tasks for the examples in Figures 1 and 2. However, it is still important for students to have a good understanding of time-value concepts, so that they can appreciate more fully the idea of bond valuation. This, in turn, provides a good foundation for valuating more sophisticated financial assets. Therefore, the reliance on bond-specific Excel functions alone, especially for price and yield determination without attention to the underlying concepts, though seemingly convenient, is inadequate from a pedagogic perspective. Nevertheless, such Excel functions are useful for verifying numerically the various analytical relationships among the underlying parameters in bond valuation.

5.1 A Different Accrual Basis

For ease of exposition, the Excel illustrations so far in this paper are for the day-counting convention based on actual calendar days. To help students learn about the practice in different bond markets, it is useful to describe briefly some common accrual methods. However, a detailed account of various accrual methods is non-essential in enhancing students' understanding of time-value concepts and bond valuation. It actually adds an extra layer of complexity in the computational tasks involved, thus tending to divert students' attention away from conceptual issues.

The use of Excel day-counting functions COUPDAYS, COUPDAYBS, COUPDAYSNC can greatly alleviate the above concern, especially when a complicated accrual basis is involved. As a detailed account of an accrual basis is non-essential in the delivery of the analytical materials in bond valuation, only a commonly used method, called 30/360, is described here. In U.S. bond markets, for example, the accrual basis for corporate bonds are for a 30-day month and a 360-day year. To implement the 30/360 accrual basis, if the settlement date or the prior coupon date is at the end of a calendar month with 28, 29, or 31 days, some specific rules have to be followed.

To illustrate, let $D_1/M_1/Y_1$ and $D_2/M_2/Y_2$ represent the prior coupon date and the settlement date, respectively. Here, D, M, and Y correspond to the day, the month, and the year involved, respectively. The specific rules are applied in the following order: (1) If both D_1 and D_2 are the last day of February, change the latter to 30; (2) if D_1 is the last day of February, change it to 30; (3) if D_1 is either 30 or 31 and D_2 is 31, change the latter to 30; (4) if D_1 is 31, change it to 30.

As an example, let us use a recent quotation of a bond issued by XTO Energy Inc., a U.S. company, with data retrieved from Bond Center of Yahoo! Finance on the internet.[14] The bond pays semi-annual coupons, for an annual amount of $5.00. According to the bond quotation, the settlement date and the maturity date are 28/04/2011 and 31/01/2015, respectively. Thus, we have $D_2 = 28$, $M_2 = 4$, and $Y_2 = 2011$. We can infer from such information that the prior coupon date is 31/01/2011, implying that $D_1 = 31$, $M_1 = 1$, and $Y_1 = 2011$.

The accrued interest is the proportion $N/360$ of the annual coupon, with $N = (D_2 - D_1) + 30(M_2 - M_1) + 360(Y_2 - Y_1)$. As the only change in D_1 or D_2 according to the above rules is to

[14] The corresponding website is http://screen.yahoo.com/bonds.html, which is freely accessible by users.

change D_1 from 31 to 30, we have $N = (28 - 30) + 30(4 - 1) + 360(2011 - 2011) = 88$. In this example, therefore, the accrued interest is $\$5.00 \times 88/360 = \$1.2222$. The accrued interest can also be computed directly via the function ACCRINT. Here, the last argument of the function is 0, indicating the use of the 30/360 accrual basis.

5.2 Odd Coupon Periods

In practice, although coupons are paid periodically over the life of a bond, there can still be exceptions pertaining to the first or last payment dates. For example, the first coupon date of a semi-annual coupon bond can be sooner than or later than six months after the issue date of the bond. Such exceptions, commonly called odd coupon periods, require the coupon payments involved to be adjusted proportionally. Each adjustment, in turn, affects the accrued interest when trading occurs, as well as price and yield computations. There are some specific Excel functions to accommodate such exceptions; they include ODDFPRICE, ODDLPRICE, ODDFYIELD, ODDLYIELD.[15] As long as students have a good understanding of time-value concepts, they should be able to follow the derivation of various algebraic relationships among the underlying parameters in bond valuation, for cases involving an odd first or final period. These Excel functions are particularly useful for verifying numerically the analytical relationships involved.

As an illustration of the above idea, let us consider the case of an odd first coupon period. Suppose that the settlement date is within this odd period.[16] Let I be the accrued interest and C_1 be the dollar amount of the first coupon. Following the same approach to reach equation (11), we can write the price of the bond at the settlement date as

$$P = \frac{1}{(1+r)^{1-\alpha}} \left[C_1 + \frac{C}{1+r} + \frac{C}{(1+r)^2} + \cdots + \frac{C}{(1+r)^{n-1}} + \frac{F}{(1+r)^{n-1}} \right] - I. \qquad (23)$$

Here, $1 - \alpha$ is the length of time between the settlement date and the first coupon date, as a proportion of a regular period. The similarity between equations (11) and (23) is readily noticeable; their only differences are in the first coupon (C or C_1) and in the accrued interest (αC or I). Thus, equation (13), which is intended for computational convenience, can easily be revised for the setting

[15] Here, the function names are self-explanatory, with "ODDF" and "ODDL" indicating odd first and last periods, respectively.

[16] For a bond that has passed its first coupon period, whether it had an odd first period is irrelevant for the purpose of its valuation. Bonds with odd last periods are less common in practice. Analytically, it is easy to accommodate such cases based on time-value concepts. Further, unless the settlement date is within an odd last period, the odd period has no effect on the determination of the accrued interest. If the settlement date is within the odd period, as there is only one remaining coupon and the face value, the corresponding valuation is analytically simple.

here. Specifically, the revised expression is

$$P = (1+r)^{\alpha} \left\{ C \left[\frac{1-(1+r)^{-n}}{r} \right] + \frac{F}{(1+r)^n} \right\} + \frac{C_1 - C}{(1+r)^{1-\alpha}} - I. \quad (24)$$

As a numerical example, let us consider the quotation of a U.S. corporate bond, with data retrieved from the afore-mentioned Yahoo! Finance website. The issuer of the bond is Health Care REIT Inc. The issue date, the settlement date, the first coupon date, and the maturity date are 14/03/2011, 28/04/2011, 15/07/2011, and 15/01/2022, respectively. For a \$100 face value, the annual coupon is \$5.25 and the price is \$103.75; the annual yield to maturity is 4.799%.

Given the 30/360 accrual basis, which pertains to U.S. corporate bond markets, although a regular coupon period is 180 days, there are only $(15-14)+30(7-3)+360(2011-2011) = 121$ interest-bearing days between the issue date and the first coupon date. Thus, the first coupon is $C_1 = \$5.25 \times 121/360 = \$1.7646$, while each of the remaining semi-annual coupons is $C = \$5.25/2 = \$2.625$. Between the issue date and the settlement date, there are $(28-14)+30(4-3)+360(2011-2011) = 44$ days apart, and thus the accrued interest is $I = \$5.25 \times 44/360 = \$0.6417$. Further, as the settlement date and the first coupon date are separated by $(15-28)+30(7-4)+360(2011-2011) = 77$ days, we have $1-\alpha = 77/180$ and thus $(C_1 - C)/(1+r)^{1-\alpha} = \$(1.7646 - 2.625)/(1.023995)^{77/180} = -0.8517$. With $r = 4.799\%/2 = 0.023995$ and $n = 22$, equation (24) leads to $P = \$103.7447$, which is consistent with the quoted bond price.

As a complementary approach, we can use the function ODDFPRICE to compute the bond price directly. The result, as expected, is also $P = \$103.7447$. The arguments of the function are the four dates as indicated earlier in this example, the annual coupon rate, the annual yield, the face value, the annual frequency of coupon payments, and the code 0 for the 30/360 accrual basis.

Notice that equation (24) can also be used to deduce various underlying parameters for bond valuation. The idea is similar to that in Section 4. Thus, the corresponding details are omitted here. The Excel function ODDFYIELD is useful for checking the numerical search result of the annual yield to maturity in the same example, if such information is concealed from the bond quotation. In this example, as expected, the numerical search result, $2r = 4.7984\%$, is consistent with that from the function ODDFYIELD.

6 Concluding Remarks

This paper has applied time-value concepts to bond valuation, from a pedagogic perspective. It intends to bridge the existing gap between the coverage of bond valuation in introductory finance textbooks and bond valuation in practice. Specifically, this paper has extended the textbook coverage by considering a more versatile model. The model allows the bond price to be stated in terms of various underlying parameters, with valuation dates not confined to coupon dates as in standard textbooks. The model considered is practical; it is able to capture the practice in bond markets that, if a bond is sold with the settlement date of the transaction not at coupon dates, the seller is entitled to receive part of the next coupon. On the one hand, it is such a practical feature that makes the analytical material involved more relevant to business students. On the other hand, it is the analytical simplicity that allows the coverage of the corresponding materials in introductory finance courses.

Electronic spreadsheets play an important pedagogic role here. Given a model for bond valuation, which relates the bond price to its various underlying parameters, each of such parameters can be solved analytically or numerically in terms of the bond price and the remaining parameters. Bond quotations, which contain some of such data and allow additional information to be deduced, can often be accessed freely from public sources. The use of Excel for the attendant computational tasks allows students to focus on the conceptual issues involved. The use of some bond-specific functions also allows students to verify their computational results. However, the use of bond-specific functions to generate some numerical results, without attention to the underlying concepts, though seemingly convenient, is inadequate from a pedagogic perspective.

Given the growing popularity of Excel, some recent editions of introductory finance textbooks have also included various Excel-based exercises for students. Such exercises are indeed valuable, provided that they are not used primarily as substitutes for learning the concepts involved. This paper, with its analytical emphasis, is intended to help students strengthen their understanding of bond valuation and the time-value concepts involved. The analytical materials here, though set at an introductory level, are expected to enable students to be better prepared for learning the valuation of more sophisticated financial assets afterwards.

References

Bodie, Z., Kane, A., and Marcus, A.J., (2004), Essentials of Investments, Fifth Edition. McGraw-Hill/Irwin.

Cahill, M.B., and Kosicki, G., (2001), “A Framework for Developing Spreadsheet Applications in Economics,” Social Science Computer Review, 19(2), 186-200.

Christie, D., (2003), SWX Swiss Exchange: Accrued Interest & Yield Calculations and Determination of Holiday Calendars, Version 03.11.2003 (http://www.six-swiss-exchange.com/download/trading/products/bonds/accrued_interest_en.pdf).

Lawrence, E.R., and Shankar, S., (2007), "A Simple and Student-Friendly Approach to the Mathematics of Bond Prices," Quarterly Journal of Business & Economics, 46(4), 91-99.

Ross, S.A., Westerfield, R.W., Jaffe, J.F., and Roberts, G.S., (2011), Corporate Finance, Sixth Canadian Edition. McGraw-Hill Ryerson.

Ross, S.A., Westerfield, R.W., Jordan, B.D., and Roberts, G.S., (2010), Fundamentals of Corporate Finance, Seventh Canadian Edition. McGraw-Hill Ryerson.

Sugden, S., and Miller, D., (2011), "Basic Finance Made Accessible in Excel 2007: 'The Big 5, Plus 2'," Spreadsheets in Education, 4(2), Article 1.